

Федеральное государственное автономное образовательное учреждение  
высшего образования “Национальный исследовательский университет ИТМО”

Факультет Программной Инженерии И Компьютерной Техники

Лабораторная работа №6

Вариант 8313

Выполнила:

Абдуллаева София Улугбековна

Группа Р3108

Проверил:

Вербовой Александр Александрович

Санкт-Петербург 2025

## Оглавление

Задание .....	3
Программа на ассемблере БЭВМ.....	3
Область представления .....	4
Область допустимых значений .....	4
Расположение данных в памяти .....	5
Методика проверки программы .....	5
Вывод .....	6

## Задание

По выданному преподавателем варианту разработать и исследовать работу комплекса программ обмена данными в режиме прерывания программы. Основная программа должна изменять содержимое заданной ячейки памяти (X), которое должно быть представлено как знаковое число. Область допустимых значений изменения X должна быть ограничена заданной функцией F(X) и конструктивными особенностями регистра данных ВУ (8-ми битное знаковое представление). Программа обработки прерывания должна выводить на ВУ модифицированное значение X в соответствии с вариантом задания, а также игнорировать все необработываемые прерывания.

Введите номер варианта

1. Основная программа должна увеличивать на 2 содержимое X (ячейки памяти с адресом 016<sub>16</sub>) в цикле.
2. Обработчик прерывания должен по нажатию кнопки готовности ВУ-1 осуществлять вывод результата вычисления функции  $F(X) = -3X + 3$  на данное ВУ, а по нажатию кнопки готовности ВУ-3 вычесть содержимое РД данного ВУ из X, результат записать в X
3. Если X оказывается вне ОДЗ при выполнении любой операции по его изменению, то необходимо в X записать минимальное по ОДЗ число.

## Программа на ассемблере БЭВМ

```
ORG 0x0                                ; происходит инициализация векторов прерывания
V0: WORD $DEFAULT, 0x180
V1: WORD $INT1, 0x180
V2: WORD $DEFAULT, 0x180
V3: WORD $INT3, 0x180
V4: WORD $DEFAULT, 0x180
V5: WORD $DEFAULT, 0x180
V6: WORD $DEFAULT, 0x180
V7: WORD $DEFAULT, 0x180

DEFAULT: IRET                          ; обработка прерывания по умолчанию

ORG 0x16
X: WORD ?
MIN: WORD 0xFFD7                       ; -41, min значение X
MAX: WORD 0x002B                       ; 43, max значение X

ORG 0x20
START: DI                              ; запрет прерываний для ВУ, которые не используются
      CLA
      OUT 0x1                          ; MR КВУ-0 на вектор 0
      OUT 0x5                          ; MR КВУ-2 на вектор 0
      OUT 0xB                          ; MR КВУ-4 на вектор 0
      OUT 0xD                          ; MR КВУ-5 на вектор 0
      OUT 0x11                         ; MR КВУ-6 на вектор 0
      OUT 0x15                         ; MR КВУ-7 на вектор 0
      OUT 0x19                         ; MR КВУ-8 на вектор 0
      OUT 0x1D                         ; MR КВУ-9 на вектор 0
      LD #0x9                          ; разрешение прерывания и вектор #1
      OUT 0x3                          ; (1000|0001) = 1001 в MR КВУ-1
      LD #0xB                          ; разрешение прерывания и вектор #3
      OUT 0x7                          ; (1000|0111) = 1011 в MR КВУ-3
      EI                              ; разрешаем прерывание

; основная программа
ORG 0x30
MAIN: LD X
      INC
      INC
      CALL CHECK                      ; проверяем, находится ли значение AC в пределах ОДЗ
      ST X
      JUMP MAIN

; проверка X на соответствие ОДЗ
ORG 0x40
CHECK: CMP $MIN
```

```

        BLT LD_MIN
        CMP $MAX
        BEQ RETURN
        BGE LD_MIN
        JUMP RETURN
LD_MIN: LD $MIN
RETURN: RET

; обработка прерывания на ВУ-1
ORG 0x50
INT1:  LD X
        CALL CHECK      ; проверяем, находится ли значение AC в пределах ОДЗ
        PUSH            ; сохранили AC
        ST X
        NOP              ; отладочная точка останова
        ASL              ; 2X
        ASL              ; 4X
        SUB X            ; 3X
        NEG              ; -3X
        ADD #0x03        ; -3X+3
        OUT 0x2          ; запись из AC по адресу в DR KBY-1
        NOP              ; отладочная точка останова
        POP              ; вернули AC назад
        EI
        IRET

; обработка прерывания на ВУ-3
ORG 0x60
INT3:  LD X
        CALL CHECK      ; проверяем, находится ли значение AC в пределах ОДЗ
        PUSH            ; сохранили AC
        ST X
        NOP              ; отладочная точка останова
        CLA
        IN 0x6
        SUB X
        CALL CHECK      ; проверяем, находится ли значение AC в пределах ОДЗ
        ST X
        NOP              ; отладочная точка останова
        POP              ; вернули AC назад
        EI
        IRET

```

## Область представления

MIN, MAX, X – знаковые 16-разрядные целые числа

DR KBY – 8-разрядное целое знаковое число

## Область допустимых значений

$$-128 \leq -3X + 3 \leq 127$$

$$-131 \leq -3X \leq 124$$

$$-124 \leq 3X \leq 131$$

$$-41 \leq X \leq 43$$

$$-41_{(10)} = 1111\ 1111\ 1101\ 0111_{(2)} = \text{FFD7}_{(16)}$$

$$43_{(10)} = 0000\ 0000\ 0010\ 1011_{(2)} = 002B_{(16)}$$

## **Расположение данных в памяти**

Векторы прерываний: 0x00 – 0x0FF

Переменные: 0x16 – 0x18

Основная программа: 0x30 – 0x38

Подпрограмма обработки для проверки ОДЗ: 0x40 – 0x47

Обработчик прерываний на ВУ-1: 0x50 – 0x5F

Обработчик прерываний на ВУ-3: 0x60 – 0x68

## **Методика проверки программы**

### **Проверка основной программы:**

1. Загрузить текст программы в БЭВМ
2. Ввести в клавишный регистр адрес 0x16, нажать “Ввод адреса”, дальше записать значение переменной X, например, можно ввести 002C, чтобы проверить, что значение X изменится на min значение (FFD7)
3. Запускаем программу в режиме “Работа”, начнётся бесконечный цикл наращивания содержимого AC, оно инкрементируется на 2 и записывается в ячейку X, если значение выйдет за границы ОДЗ, то в X запишется min значение

### **Проверка обработки прерываний:**

1. Заменить NOP на HLT в точках останова
2. Загрузить текст программы в БЭВМ
3. Ввести в клавишный регистр адрес 0x16, нажать “Ввод адреса”, записать значение переменной X, дальше вернуть счётчик команд в исходное состояние (ввести адрес 0x20)
4. Запустить программу в режиме “Работа”
5. Установить готовность ВУ-1
6. Дождаться остановки программы
7. Нажать “Продолжить”
8. Рассчитать ожидаемый результат вычисления значения функции  $F(X) = -3X + 3$  и сравнить его с содержимым AC
9. Дождаться остановки программы
10. Результат обработки прерывания (значение AC) запишется в DR KBY-1
11. Ввести в DR KBY-3 произвольное число и установить готовность ВУ-3
12. Нажать “Продолжить”

13. Дождаться остановки программы
14. Нажать “Продолжить”
15. Из значения АС вычитается X, новое значение проверяется на соответствие ОДЗ и записывается в ячейку X

Прерывание на ВУ-1			
АС (значение X)	Значение после ОДЗ	Ожидание $-3*X+3$	DR KBY-1
002B <sub>(16)</sub> (43)	002B <sub>(16)</sub> (43)	FF82 <sub>(16)</sub> (-126)	82 <sub>(16)</sub> (-126)
FFD6 <sub>(16)</sub> (-42)	FFD7 <sub>(16)</sub> (-41)	007E <sub>(16)</sub> (126)	7E <sub>(16)</sub> (126)
0012 <sub>(16)</sub> (18)	0012 <sub>(16)</sub> (18)	FFCD <sub>(16)</sub> (-51)	CD <sub>(16)</sub> (-51)

Прерывание на ВУ-3				
АС (значение X)	Значение после ОДЗ	DR KBY-3	Результат (DR-AC)	Результат АС (значение X)
0028 <sub>(16)</sub> (40)	0028 <sub>(16)</sub> (40)	32 <sub>(16)</sub> (50)	000A <sub>(16)</sub> (10)	000A <sub>(16)</sub> (10)
FFFF <sub>(16)</sub> (-1)	FFFF <sub>(16)</sub> (-1)	1D <sub>(16)</sub> (29)	001E <sub>(16)</sub> (30)	001E <sub>(16)</sub> (30)
0023 <sub>(16)</sub> (35)	0023 <sub>(16)</sub> (35)	28 <sub>(16)</sub> (40)	0005 <sub>(16)</sub> (5)	0005 <sub>(16)</sub> (5)
FFD4 <sub>(16)</sub> (-44)	FFD7 <sub>(16)</sub> (-41)	08 <sub>(16)</sub> (8)	0031 <sub>(16)</sub> (49)	FFD7 <sub>(16)</sub> (-41)

## Вывод

В процессе выполнения лабораторной работы я узнала, как устроен процесс прерывания в БЭВМ, изучила команды работы разрешения и запрещения прерываний БЭВМ, программного прерывания и возврата из него. Также поработала с ВУ-1 и ВУ-3 в режиме прерывания программы.